
4. Spôsoby adresovania

Keď hovoríme o spôsoboch adresovania, máme na mysli rôzne spôsoby, ako sa v inštrukciách odvolávame na operandy v registroch alebo v pamäti.

Implicitné adresovanie

Pri tomto spôsobe adresovania operand v zápise inštrukcie nevidíme. Ako príklad môžeme uviesť inštrukciu mul bl, ktorá vynásobí obsah registra AL obsahom registra BL a výsledok uloží do registra AX. V inštrukcii mul bl sú uvedené dve implicitné adresy: adresa 1. operandu (register AL) a adresa výsledku (register AX). Žiadna z týchto adries nie je explicitne zapísaná v inštrukcii. Sú dané typom inštrukcie a schopnosťami procesora. Všetky ďalej uvádzané typy adries sú adresy explicitné.

Priamy operand (adresa 0. rádu)    

Hodnota operandu je zapísaná priamo v inštrukcii, nejde vlastne ani o adresu (preto adresa 0. rádu). Napr. v inštrukcii mov al,3 je číslo 3 priamym operandom.

Priama adresa (adresa 1. rádu)

Priama adresa sa používa najčastejšie. Je to v inštrukcii zapísané označenie registra alebo číslo pamäťového miesta, s ktorého obsahom inštrukcia pracuje. Príkladom môže byť inštrukcia add dx,[Alfa], ktorá sčíta obsah registra DX s obsahom premennej Alfa a výsledok uloží do DX. Obidva operandy inštrukcie sú priamou adresou. Meno premennej interpretuje prekladač ako offset v dátovom segmente, ktorého bázová adresa je v registri DS. Hranaté zátvorky uzatvárajúce meno premennej nie sú povinné, majú len zdôrazniť, že pracujeme s hodnotou premennej, a nie s jej adresou.

Nepriama adresa (adresa 2. rádu)

Pri nepriamom adresovaní je v inštrukcii zapísaná adresa pamäťového miesta, na ktorom nájdeme adresu operandu (na tomto mieste nájdeme priamu adresu). Nepriame adresovanie v takejto podobe sa používa zriedka (napr. v inštrukciách nepriameho skoku – pozri kapitolu 8.5). 

Za zvláštny prípad nepriameho adresovania môžeme považovať nepriame adresovanie pomocou registra. V tomto prípade je priama adresa uložená v niektorom registri procesora a inštrukcia obsahuje adresu (meno) registra. Priamu adresu môžeme získať aj sčítaním obsahov dvoch registrov, prípadne k nim môžeme ešte pripočítať číslo uvedené v inštrukcii (tzv. posunutie). Priamu adresu môžeme vypočítať pomocou 

· bázového registra,

· indexového registra,

· bázového registra a posunutia,

· indexového registra a posunutia,

· bázového a indexového registra,

· bázového registra, indexového registra a posunutia.

Procesory i80x86 umožňujú v 16-bitovom režime používať dva bázové registre: BX a BP. Ak použijeme v inštrukcii bázový register BX, chápe sa jeho obsah ako offset v dátovom segmente (adresovanom pomocou DS), na ktorom nájdeme operand. Napr. inštrukcia mov ax,[bx] skopíruje do registra AX obsah slova na adrese DS:BX. Ak sa použije bázový register BP, bude sa adresa operandu hľadať v zásobníkovom segmente (adresovanom pomocou SS). Nepriamu adresu môžeme vytvoriť aj tak, že v inštrukcii uvedieme bázový register a posunutie. Napr. príkaz mov dx,[bp+2] skopíruje do registra DX obsah slova v zásobníkovom segmente. Offset slova (priamu adresu) vypočítame tak, že k obsahu registra BP pripočítame 2. Posunutím môže byť aj konštanta alebo premenná. V prípade premennej sa k obsahu bázového registra pripočíta adresa premennej.

V 16-bitovom režime sú dostupné dva indexové registre: SI a DI. Pri výpočte priamej adresy pomocou indexových registrov sa postupuje rovnako ako pri bázovom registri BX – obsah SI, resp. DI sa chápe ako offset v dátovom segmente, na ktorom je uložený operand. Rovnaká je aj syntax, napr. mov ax,[di]. 
Ako sme už uviedli, môžeme používať aj kombináciu bázových a indexových registrov, napr. mov al,[bx+di]. Obsahy bázového a indexového registra sa sčítajú a dostaneme offset, na ktorom budeme hľadať pravý operand. Segment sa určí podľa použitého bázového registra, t.j. v tomto prípade by sa použil dátový segment. V príkaze mov ax,[bp+si+4] sa odkazujeme do zásobníkového segmentu. Tento príklad zároveň ukazuje, že k bázovému a indexovému registru môžeme pripočítať aj posunutie. Nie je dovolené použiť súčasne dva bázové alebo dva indexové registre. 

Nepriame registrové adresovanie sa s výhodou používa pri spracovaní polí dát. Do bázového registra sa ukladá adresa začiatku poľa (báza poľa), zatiaľ čo indexový register slúži na prechádzanie poľom. Ukážme si príklad na použitie bázových a indexových registrov.

Úloha 4.1

Vypíšte na obrazovku znakový reťazec uložený v premennej Retaz a ukončený nulou.

Riešenie

Každý znak reťazca je v pamäti uložený v jednom bajte v kóde ASCII. Reťazec je uložený zľava doprava, t.j. prvý znak má najnižšiu adresu. Meno premennej interpretuje prekladač ako adresu, na ktorej reťazec začína. Na obr. 4.1 je znázornené uloženie reťazca „No nazdar!“. Kvôli lepšej čitateľnosti sú v pamäťových bunkách zapísané znaky a nie ich ASCII kódy. 




Retaz     :
’N’

Retaz +  1:
’o’

Retaz +  2:
’ ’

Retaz +  3:
’n’

(
(




Retaz +  9:
’!’

Retaz + 10:
0




Obr. 4.1 Uloženie reťazca v pamäti

Z cvičných dôvodov budeme reťazec vypisovať po jednotlivých znakoch. Nasledujúci program rieši našu úlohu. Komentované sú len inštrukcie potrebné pre pochopenie funkcie bázových a indexových registrov.

.MODEL small

.STACK 100h

.DATA

Retaz DB ’No nazdar!’,0 ; deklarácia premennej

.CODE

Zac:
mov ax,@data


mov ds,ax ; ulož bázovú adresu dátového segmentu do ds


mov bx,offset Retaz ; ulož do bx adr. 1.znaku reťazca 

Vypis:

mov di,0       ; 1. znak má index 0

VypisZnak:



mov dl,[bx+di] ; ulož do dl znak na adrese [ds]:[bx+di]
cmp dl,0       ; porovnaj hodnotu v dl s 0


jz Koniec      ; ak bola 0, choď na návestie Koniec


mov ah,2


int 21h ; vypíš na obrazovku znak, ktorého kód je v dl


inc di         ; zvýš index o 1 - prejdi na ďalší znak


jmp VypisZnak

Koniec:


mov ax,4C00h


int 21h

END Zac

Bez nepriameho adresovania by sme cyklus na výpis reťazca po znakoch ani nevedeli zostaviť. Pri výpise jedného reťazca by sme vystačili s indexovým registrom. Príkaz na skopírovanie znaku s indexom DI do registra DL by potom bol mov dl,[Retaz+di]. Ak by sme ale potrebovali vypísať viacej reťazcov, mohli by sme časť programu medzi návestiami Vypis a Koniec zapísať ako procedúru. Pred jej vyvolaním z hlavného programu by sme uložili adresu požadovaného reťazca do registra BX.

Nepriame registrové adresovanie je podporované aj pravidlami kódovania inštrukcií. Pokiaľ inštrukcia obsahuje len bázový a/alebo indexový register (bez posunutia), potrebujeme na jej zakódovanie o 2 bajty menej, ako keby sme v inštrukcii uviedli priamu adresu. 

V 32-bitovom režime sú možnosti tvorby nepriamej adresy pomocou registra oveľa bohatšie. Ktorýkoľvek 32-bitový univerzálny register môže byť použitý ako bázový register alebo (s výnimkou registra ESP) ako indexový register. Príklad: 

mov dx,[eax+edi+2]

Indexový register sa môže násobiť číslom 2, 4 alebo 8, čo umožňuje zrýchliť prístup k položkám polí typu word, dword, resp. qword (pozri kapitolu 6). Príklad:

mov eax,[DwordTable+ebx(4]


Rovnaký register môže dokonca byť použitý dvakrát: ako bázový aj ako indexový register. Táto možnosť sa dá využiť aj v 16-bitovom režime v niektorých inštrukciách, ktoré nepristupujú do pamäti, napr. inštrukcia

lea eax,[eax+eax(4]

vynásobí obsah registra EAX piatimi (pozri kapitolu 8.1 a 8.2).

Segment operandu sa určuje podľa bázového registra: ak je bázovým registrom EAX, EBX, ECX, EDX, ESI, EDI, operand leží v dátovom segmente. Ak je bázovým registrom EBP alebo ESP, operand leží v zásobníkovom segmente. Pretože ľubovoľný univerzálny register môže byť bázou aj indexom, otázka je, za čo ho bude prekladač považovať a ktorý segmentový register sa pri výpočte adresy použije. Pravidlá sú nasledujúce: Ak sa v odkaze vyskytujú dva registre, len jeden z nich sa môže násobiť, a ten sa potom považuje za indexový register; druhý register bude bázový. Ak nie je použité násobenie, ľavý register bude bázový, pravý indexový. Ak je použitý len jeden register, považuje sa za bázový bez ohľadu na to, či ho násobíme konštantou alebo nie.  























































































































































































































































































































































