
5. Základné prvky assemblerovského programu

Inštrukcie v zdrojovom kóde píšeme po jednej na riadok. Syntax riadku je nasledujúca:

[identifikátor]  inštrukcia/direktíva [operandy] [;poznámka]

Hranaté zátvorky označujú prvok, ktorý je nepovinný (nemusí byť v každom riadku), lomka oddeľuje alternatívy. 

Identifikátorom rozumieme symbolické meno konštanty alebo miesta v pamäti, t.j. symbolickú adresu inštrukcie alebo dátového objektu (presnejšie offset v príslušnom segmente). Identifikátor môže obsahovať písmená, číslice a znaky . , _ , ? , $ , @ . Identifikátor nemôže začínať číslicou. Nerozlišujú sa malé a veľké písmená. Ako bolo vidieť v príklade v predchádzajúcej kapitole, zdrojový kód v JSA nie je veľmi čitateľný. Preto na tomto mieste zdôrazňujeme: Píšte samodokumentujúce identifikátory! Aj v pekne štruktúrovanom pascalovskom programe sa dá lepšie zorientovať, keď mená premenných súvisia s ich obsahom. V  assemblerovskom programe, ktorý je neprehľadný už sám osebe, sa bez čitateľných mien premenných a návestí stratíme. Na ilustráciu uvedieme krátku programovú sekvenciu, ktorá má zistiť, či v registri AL je malé písmeno. Ak je, treba ho previesť na veľké písmeno. Pri tomto prevode sa využíva skutočnosť, že kódy písmen v systéme ASCII nasledujú za sebou v stúpajúcej postupnosti a že kód veľkého písmena je o hodnotu 20h menší než kód zodpovedajúceho malého písmena. Test na malé písmeno sa robí tak, že porovnáme obsah registra AL s kódmi znakov „a“ a „z“ (inštrukcia cmp). Ak je v registri AL číslo menšie ako kód „a“ alebo väčšie ako kód písmena „z“, znamená to, že v AL nie je malé písmeno. V opačnom prípade odčítame od AL 20h (inštrukcia sub), a tak vykonáme konverziu malého písmena na veľké. Lenivý programátor by program napísal takto:

cmp al,’a’ 

jb N1

cmp al,’z’

ja N1

sub al,20h

N1: 
(
Určite prehľadnejšie však vyzerá program:

cmp al,’a’ 
jb NieJeMalePismeno 

cmp al,’z’ 

ja NieJeMalePismeno 

sub al,20h

NieJeMalePismeno:

 
(
Druhou položkou v riadku je inštrukcia alebo direktíva. Inštrukcie budú podrobne popísané v kapitole 8. Stretneme sa aj s direktívami, ale nevymenujeme ich všetky, pretože ich je veľa, sú uvedené v inej literatúre (napr. [1]) a väčšinu z nich pravdepodobne nikdy nepoužijete. Nateraz vám stačí vedieť, že direktívy sa na rozdiel od inštrukcií neprekladajú do strojového kódu. Slúžia na:

1. definovanie konštánt, premenných, návestí, segmentov, procedúr a makier;

2. aktivovanie inštrukčného súboru procesorov 80186, 80286, 80386, 80486, Pentium a koprocesorov 80287 a 80387;

3. definovanie tvaru a obsahu protokolu o preklade;

4. podmienený preklad (napr. direktívy IF, ENDIF, ELSE).

Operandami inštrukcií môžu byť konštanty, registre alebo adresy pamäťových miest. To, ako sa operandy sprístupňujú, sme uviedli v predchádzajúcej kapitole venovanej spôsobom adresovania. Pristavme sa pri zápise konštánt. Konštantami môžu byť celé alebo reálne čísla, znaky a znakové reťazce. Celé čísla sa udávajú v sústave desiatkovej, dvojkovej alebo šestnástkovej. Zápis čísla v sústave dvojkovej končí písmenom „b“. Zápis čísla v šestnástkovej sústave musí začínať číslicou, aby prekladač odlíšil zápis čísla od identifikátora. Zápis čísla v šestnástkovej sústave končí písmenom „h“ (napr. 0FEh). Reálne čísla sa zapisujú v sústave desiatkovej, a to buď v tvare s desatinnou bodkou (napr. 1.5) alebo v exponenciálnom tvare (napr. 1.32E1). Môžu sa tiež zapísať v šestnástkovej sústave podľa normy IEEE, musia však obsahovať všetky číslice príslušného formátu a končiť príponou „r“ , napr. 0C014CCCCCCCCCCCCr je číslo -5.2 uložené v 8 bajtoch. Postupnosť znakov v zápise znakovej alebo reťazcovej konštanty musí byť uzavretá v apostrofoch alebo v úvodzovkách.

Inštrukcia nemusí mať žiadny operand, môže mať jeden, dva alebo tri operandy. Inštrukcie pre presun údajov, aritmetické a logické inštrukcie majú dva operandy. Ktorý z nich je zdroj a ktorý cieľ? Nad ktorým sa vykoná operácia a kam sa uloží výsledok? (Ak napíšem mov ax,bx, skopíruje sa obsah BX do AX alebo naopak?) Zapamätajme si nasledujúce pravidlo: ak má inštrukcia dva operandy, pravý je zdroj (s ním sa vykonáva príslušná operácia) a ľavý môže byť:

· zdroj, ak inštrukcia požaduje dva operandy, ale výsledok operácie nikam neukladá, napr. cmp al,1 porovná  obsah registra AL s hodnotou 1 tak, že odčíta 1 od AL, nastaví príslušné príznakové bity, ale výsledok odčítania nikam neuloží,

· cieľ – napr. známa inštrukcia mov uloží pravý operand do ľavého operandu. Pre lepšie zapamätanie si predstavme analógiu s Pascalom: mov al,0 ( al := 0,

· zdroj a cieľ súčasne, napr. add al,bl sčíta obsah registra AL s obsahom registra BL a výsledok uloží do AL (v Pascale al := al + bl).

Ak je jedným z operandov odkaz na pamäťové miesto, druhým musí byť register alebo konštanta. S výnimkou reťazcových inštrukcií nie je dovolená operácia typu pamäť – pamäť. Konštanta nemôže byť ľavým operandom. Ak sú operandami registre alebo pamäťové miesta, musia sa typy ľavého a pravého operandu zhodovať, t.j. obidva operandy musia mať veľkosť bajtu, slova alebo dvojslova (32 bitov). Nemožno napísať napr. mov al,bx. Ako uvidíme v kapitole 8, existujú zriedkavé výnimky z tohto pravidla, napr. inštrukcia movzx.  

Poznámky uľahčujú orientáciu v programe, čo oceníme najmä pri ladení programu a pri jeho neskorších úpravách.

6. Premenné a návestia

6.1 Premenné

Premenné definujeme pomocou direktív DB, DW, DD, DF, DP, DQ a DT, ktoré vyhradzujú premenným miesto v pamäti a zároveň im prideľujú hodnotu. Deklarácia premenných má tvar, ktorý zodpovedá formátu riadku, ako bol uvedený v predchádzajúcej kapitole:

[meno_premennej]  Dx výraz [,výraz] (,

kde x je jedno z písmen B, W, D, F, P, Q, T. Direktívy prideľujú pamäť v jednotkách, ktoré sú určené druhým písmenom direktívy (pozri tab. 6.1). Zároveň direktíva určuje typ premennej. Jedna direktíva môže prideliť jednu alebo viacej jednotiek (ak za Dx nasleduje viacej výrazov). Ak nie je uvedené meno premennej, direktíva len inicializuje časť pamäti, t.j. uloží do pamäti dané hodnoty, ale adresu pamäti nepriradí žiadnej premennej. Výrazom môže byť jeden z nasledujúcich prvkov:

a) číselný výraz obsahujúci konštanty, 

b) adresový výraz, 

c) znakový výraz,

d) znak ? – direktíva pridelí premennej jednotku pamäti, ale neinicializuje ju (obsah slabiky bude nedefinovaný),

e) konštrukcia 

n dup (výraz [,výraz] ()

pre n-násobnú inicializáciu rovnakou hodnotou uvedenou v zátvorke. Výrazom môže byť prvok a), d) alebo e).

Tabuľka 6.1

Direktívy definujúce premenné

Direktíva
Veľkosť pridelenej pamäti v bajtoch
Typ premennej
Do premennej môžeme uložiť

DB
1
byte
Celé číslo so znamienkom v rozsahu (-128; 127(
Celé číslo bez znamienka v rozsahu (0; 255(
Znak 

DW
2
word
Celé číslo so znamienkom v rozsahu (-32 768; 32 767(
Celé číslo bez znamienka v rozsahu (0; 65 535(
16-bitový offset

DD
4
dword
Celé číslo so znamienkom v rozsahu (-231; 231-1(
Celé číslo bez znamienka v rozsahu (0; 232-1(
Reálne číslo v rozsahu (1.5E-45; 3.4E38(

Far pointer v 16-bitovom režime, t.j. adresa segment:offset (16-bitový)

32-bitový offset

DP

DF
6 
pword

fword
Celé číslo so znamienkom v rozsahu (-247; 247-1(
Celé číslo bez znamienka v rozsahu (0; 248-1(
Far pointer v 32-bitovom režime, t.j. adresa segment:offset (32-bitový)

DQ
8
qword
Celé číslo so znamienkom v rozsahu (-263; 263-1(
Celé číslo bez znamienka v rozsahu (0; 264-1(
Reálne číslo v rozsahu (5.0E-324; 1.7E308(

DT
10
tbyte
Celé číslo so znamienkom v rozsahu (-279; 279-1(
Celé číslo bez znamienka v rozsahu (0; 280-1(
Celé číslo v kóde BCD 

Reálne číslo v rozsahu (3.4E-4932; 1.1E4932(

Nasledujú príklady jednotlivých direktív. Prvá direktíva ORG 100h spôsobí, že nasledujúce dáta sa budú ukladať od offsetu 100h v danom segmente. Všimnite si, že čísla sa do pamäti ukladajú v obrátenom poradí slabík. Preto adresa premennej ukazuje na najnižšiu slabiku čísla a vyššie slabiky čísla sa ukladajú postupne do slabík pamäti s vyššími adresami. Direktíva DT uloží desiatkové číslo v zhustenom formáte BCD (každú číslicu zobrazí v 4 bitoch). Záporné číslo BCD má v najvýznamnejšom bite hodnotu 1.

Príklady

      ORG 100h             ; obsahy slabík od offsetu 100h:

Jedna DB –1,12+1           ; FF 0D
      DB ’abcd’            ; 61 62 63 64

      DB 2 dup(3 dup(?),2) ; ?? ?? ?? 02 ?? ?? ?? 02

      DW –32768            ; 00 80

      DW Jedna             ; 00 01

      DD 0.1               ; CC CC CC 3D

      DD 0ABCDEF23h        ; 23 EF CD AB

      DD Jedna             ; 00 01 xy uv

      DT 1234567890        ; 90 78 56 34 12 00 00 00 00 00

      DT –1234567890       ; 90 78 56 34 12 00 00 00 00 80

6.2 Návestia

Návestia identifikujú miesto v programe, na ktoré sa chceme odvolávať. Sú to vlastne symbolické adresy inštrukcií. V zmysle tejto definície považujeme za návestie aj meno procedúry. Návestie môže byť buď typu near (blízky) alebo far (vzdialený). 

Návestie má typ near vtedy, keď sa naň budeme odvolávať z rovnakého segmentu, v ktorom je návestie definované. Návestie typu near sa definuje pomocou direktívy : , pričom identifikátor návestia nasledovaný dvojbodkou je na samostatnom riadku, alebo je za ním na rovnakom riadku uvedená inštrukcia, ku ktorej sa návestie vzťahuje, napr.

Cyklus:

      mov ax,bx

 Okrem toho môžeme návestie typu near definovať pomocou direktívy LABEL, napr.

Cyklus LABEL near

      mov ax,bx

alebo direktívou PROC so špecifikáciou near:

Faktorial PROC near

Ak v direktíve PROC nie je uvedená špecifikácia, návestie má typ near.

Keď sa odkaz na návestie nachádza v inom segmente, musí mať návestie typ far. Návestie typu far definujeme pomocou direktívy LABEL, napr.

Urob LABEL far

      mov ax,bx

Návestie typu far vo funkcii vstupného bodu procedúry definujeme direktívou PROC:

Vypis PROC far

6.3 Atribúty premenných a návestí

Premenné a návestia sú charakterizované atribútmi

· segment – odpovedá bázovej adrese segmentu, v ktorom je definovaná premenná (návestie);
· offset premennej (návestia) v segmente, v ktorom je definovaná;
· typ.

Vo výrazoch môžeme používať operátory, ktoré vracajú alebo prepisujú hodnoty týchto atribútov.

6.3.1 Operátory vracajúce hodnoty atribútov

Operátory vracajúce hodnoty atribútov sú zhrnuté v tab. 6.2.

Tabuľka 6.2

Operátory vracajúce hodnoty atribútov

Operátor
Význam

seg
Vráti hodnotu atribútu segment

offset
Vráti hodnotu atribútu offset

type
Vráti hodnotu
podľa typu premennej (návestia)


1

2

4

6

6

8

10

-1

-2
byte

word

dword

fword

pword

qword

tbyte

near

far

length
Vráti počet jednotiek pamäti pridelených premennej

size
Vráti length(type

Ako vidieť z tabuľky, operátor type v spojení s premennou vráti počet bajtov, ktoré zaberá jednotka pamäti pridelená premennej direktívou Dx. Hodnota, ktorú vráti operátor, sa nemení, aj keď je premennej pridelených viacej jednotiek (ak je premennou povedzme vektor alebo reťazec). 

Operátor length vráti hodnotu väčšiu ako 1 len v prípade viacnásobnej deklarácie s použitím operátora dup.

Príklady

Hodnota DW 1234h

Post DB 5,6,7

Pole DW 100 dup(0) 

mov al,type Hodnota   ; al := 2

mov bl,type Post      ; bl := 1

mov cl,type Pole      ; cl := 2

mov ah,length Hodnota ; ah := 1

mov bh,length Post    ; bh := 1

mov ch,length Pole    ; ch := 100

mov dx,size Pole      ; dx := 200

6.3.2 Operátory prepisujúce hodnoty atribútov

Operátory prepisujúce hodnoty atribútov sú zhrnuté v tab. 6.3.

Tabuľka 6.3

Operátory prepisujúce hodnoty atribútov

Operátor
Prepisuje atribút

:
segment

ptr
typ

Operátor prepisu segmentu umožňuje zmeniť predpokladaný segment pamäťovej referencie. Napr. ak je odkazom meno premennej, prekladač ho implicitne chápe ako offset v dátovom segmente. Keď chceme, aby inštrukcia pristupovala do iného segmentu, než je štandardne nastavený, napíšeme pred odkaz iný segmentový register s operátorom prepisu segmentu, napr.

mov al,ss:[Znak]

V tab. 6.4 sú zhrnuté štandardné a alternatívne segmentové registre pre rôzne typy pamäťových referencií.

Pred operátorom prepisu segmentu môžeme uviesť aj meno segmentu – pozri kapitolu 9.

Tabuľka 6.4

Štandardné a alternatívne segmentové registre

Typ pamäťovej referencie
Štandardný segmentový register
Alternatívny segmentový register

Návestie
CS
(

Priama adresa, nepriama adresa s bázou BX a/alebo indexom SI, DI, nepriama adresa s bázou EAX, EBX, ECX, EDX, ESI, EDI
DS
CS, ES, FS, GS, SS

Nepriama adresa s bázou BP, nepriama adresa s bázou EBP, ESP
SS
CS, DS, ES, FS, GS

Zdrojový reťazec
DS
CS, ES, FS, GS, SS

Cieľový reťazec
ES
(

Operátor ptr dovoľuje napr. pristupovať k jednotlivým bajtom premennej typu word, alebo naopak, považovať samostatne definované bajty za jedno slovo: 

Hodnota DW 1234h

Cislo DB 1,2

mov al,byte ptr [Hodnota] ; al := 34h

mov bx,word ptr [Cislo]   ; bx := 201h

Operátor ptr je nevyhnutný v situáciách, kedy prekladač nevie posúdiť, s akým typom pamäťových operandov inštrukcia pracuje. Napr. v inštrukcii 

mov [bx],1

nie je jasné. či chceme uložiť 1 do bajtu alebo slova, ktorého adresa je v registri BX. Takúto inštrukciu prekladač považuje za chybnú a vyžaduje špecifikáciu typu pamäťového operandu:

mov byte ptr [bx],1

alebo

mov word ptr [bx],1

Medzi operátory prepisujúce hodnoty atribútov by sme mohli zaradiť aj operátor this, ktorý definuje premennú alebo návestie daného typu na danom mieste programu. Používa sa v spojení s direktívou EQU vo forme príkazu

identifikátor EQU this typ

Pomocou tohto operátora môžeme pristupovať k už definovanému objektu pod iným menom s iným typom bez toho, aby sme v inštrukcii museli použiť prepis typu, napr.

Hodnota EQU this word

Cislo DB 1,2

mov al,[Cislo]   ; al := 1

mov bx,[Hodnota] ; bx := 201h

Podobne v kódovom segmente môžeme jednu inštrukciu označiť dvomi návestiami, pričom jedno bude mať typ near a budeme sa naň odkazovať z toho istého segmentu, druhému priradíme typ far a budeme ho používať v odkazoch z iného segmentu, napr.:

Nav_far EQU this far

Nav_near: mov ax,0FFFFh

7. Symbolické konštanty

Podobne ako vo vyššom programovacom jazyku, aj v JSA sa symbolické mená pre konštanty zavádzajú preto, aby uľahčili orientáciu v programe, prípadne jeho modifikáciu. Určite dá menej námahy opraviť konštantu v programe raz, než keby sme museli prezerať celý program a opravovať každý výskyt. Konštanty sa definujú pomocou direktívy EQU alebo direktívy =. Rozdiel medzi oboma direktívami je ten, že konštantu zavedenú pomocou direktívy = možno predefinovať. 

Symbolické konštanty často využívajú preddefinovaný symbol $. Konkrétna hodnota sa symbolu $ prideľuje pri preklade a je ňou aktuálny offset v danom segmente. V nasledujúcom príklade symbol $ reprezentuje adresu slabiky v dátovom segmente, ktorá nasleduje za reťazcovou premennou Retaz.

Úloha 7.1

Vypíšte na obrazovku znakový reťazec uložený v premennej Retaz. 

Riešenie

Táto úloha je modifikáciou úlohy 4.1, ale reťazec teraz nie je ukončený nulou. Pretože ho opäť chceme vypisovať po jednotlivých znakoch, využijeme symbol $ na získanie dĺžky reťazca. Dĺžku reťazca uložíme do konštanty PocetZnakov.

.MODEL small

.STACK 100h

.DATA

Retaz DB ’No nazdar!’

PocetZnakov EQU $-Retaz 

.CODE

Zac:
mov ax,@data


mov ds,ax ; ulož bázovú adresu dátového segmentu do ds


mov bx,offset Retaz ; ulož do bx adr. 1.znaku reťazca 

      mov cx,PocetZnakov  ; počet opakovaní cyklu VypisZnak
Vypis:

mov di,0            ; 1. znak má index 0

VypisZnak:



mov dl,[bx+di] ; ulož do dl znak na adrese [ds]:[bx+di]

mov ah,2


int 21h ; vypíš na obrazovku znak, ktorého kód je v dl


inc di ; zvýš index o 1 - prejdi na ďalší znak


loop VypisZnak ; opakuj výpis 


mov ax,4C00h


int 21h

END Zac























































































































































































































































































































































� Reálne čísla môžu byť kladné aj záporné; rozsah platí pre absolútnu hodnotu čísla. 


� Reálne čísla sa ukladajú v štandardnom formáte IEEE.


� uvxy je bázová adresa dátového segmentu; 0100h je offset premennej Jedna nastavený direktívou ORG 100h.


� Direktíva EQU v spojení s operátorom this má rovnaký význam ako direktíva LABEL, tzn. uvedený zápis je ekvivalentný so zápisom: identifikátor LABEL typ.





